From: Steve.Hayes@p5.f22.n7101.z5.fidonet.org (Steve Hayes)
Newsgroups: soc.roots
Subject: PAF to Paradox
Date: Sat, 05 Sep 92 20:40:00 GMT
Organization: FidoNet node 5:7101/22.5 - Novell User Group, Pretoria RSA

I have almost succeeded in creating a script to convert data from PAF 
INDIV.DAT to Paradox via Steve Cannon's FRU program, which creates a 
comma-delimited Ascii file from INDIV2.DAT.

There is only one thing wrong - I haven't found what information Field 36 - 
the second last field - contains. If anyone knows, please tell me. But in the 
meantime, this script should work. If you have PAF, Paradox and FRU, please 
try it. You can copy it from this message. I suggest you name it FR_IMP.SC, 
but you can call it anything you like as long as it ends in .SC.

======== cut here ========

; ====== FR_IMP.SC
; ====== Script to import INDIV2.DAT PAF data into Paradox via FRU
;               Version 1.0 by Steve Hayes dated 1992-09-05
;               Fidonet: steve.hayes of 5:7101/22
;               Internet: steve.hayes@f22.n7101.z5.fidonet.org
ClearAll
Clear
release procs all
release vars all
fileans = ""
Text
This script will import data from the INDIV2.DAT file  of the Personal
Ancestral File (PAF) program into a Paradox table.

Before you can import the file into Paradox, you will need to create a
comma-delimited Ascii file from the PAF data using the Family Records
Utilities program (FRU) by Steve Cannon. FRU will create a file called
FR.IND (Family Records - Individual). This file needs to be present before
this script will work.
Endtext
Sleep 5000

PROC GetTable()
  @ 10,0
  Clear EOS
  @ 10, 10
  ? "Enter the name of the table to import the data to: "
  accept "a8" picture "!*!" to fileans
ENDPROC

while isblank(fileans)
  GetTable()
  if not isblank(fileans)
    then
    quitloop
  endif
  @ 12,10
  ? "You must enter a name for the file"
  Sleep 3000
endwhile

If isfile(fileans+".db")
  then
    beep
    @ 10,0
    Clear EOS
    @ 10,10
      ? fileans, " already exists."
      ?
      ? "The old file will be renamed to ZZFILE"
      ? "Please rename or delete ZZFILE after ", fileans, " has been created"
      if not istable("zzfile")
        then
        rename fileans "zzfile"
        else
        Clear
        @ 10,10
        ? "You didn't rename ZZFILE! Please do it now"
        @ 12,10
        ? "Script aborting"
        sleep 5000
        quit
        endif
      sleep 5000
Endif

?
? "Enter the drive and directory in which the FR.IND file is located "
accept "a20" picture "!*!" to path
if substr(path,(len(path)),1) <> "\\"
  then path = path + "\\"
endif
infile = path + "FR.IND"
if not isfile(infile)
  then
    clear
    @ 10,10
    ? "FR.IND not found in path."
    @ 12,10
    ? "Script aborting - please create FR.IND with the FRU program"
    Sleep 5000
    Quit
Endif

@ 10,0
Clear EOS
@ 10,10
? "Creating ", fileans
create fileans
  "Rin"                  : "S",
  "Surname"              : "A16",
  "Given name 1"         : "A16",
  "Given name 2"         : "A16",
  "Given name 3"         : "A16",
  "Title"                : "A6",
  "Sex"                  : "A1",
  "Birth date"           : "A15",
  "Birth place 1"        : "A16",
  "Birth place 2"        : "A16",
  "Birth place 3"        : "A16",
  "Birth place 4"        : "A16",
  "Baptism date"         : "A15",
  "Baptism place 1"      : "A16",
  "Baptism place 2"      : "A16",
  "Baptism place 3"      : "A16",
  "Baptism place 4"      : "A16",
  "Death date"           : "A15",
  "Death place 1"        : "A16",
  "Death place 2"        : "A16",
  "Death place 3"        : "A16",
  "Death place 4"        : "A16",
  "Burial date"          : "A15",
  "Burial place 1"       : "A16",
  "Burial place 2"       : "A16",
  "Burial place 3"       : "A16",
  "Burial place 4"       : "A16",
  "LDS Baptism date"     : "A15",
  "LDS Baptism temple"   : "A3",
  "LDS Endowment date"   : "A15",
  "LDS endowment temple" : "A3",
  "LDS sealing date"     : "A15",
  "LDS sealing temple"   : "A3",
  "Sibling no"           : "A4",
  "Mrin"                 : "S",
  "Unknown"              : "S",
  "Id number"            : "A10"
? fileans, " created. Now importing data"
{Tools} {ExportImport} {Import}
{Ascii} {AppendDelimited}
select infile
select fileans
release procs all
release vars all
========== cut here =======

Steve Hayes
P O Box 7648, Pretoria
0001 South Africa



--  
INTERNET: Steve.Hayes@p5.f22.n7101.z5.fidonet.org
via:  THE CATALYST BBS in Port Elizabeth, South Africa.
       (catpe.alt.za)   +27-41-34-1122 HST or +27-41-34-2859, V32bis & HST.
Comments: Gated by NETNEWS@AUVM.AMERICAN.EDU